

```

> # Input: x = [x1, x2, x3]
# q is an irreducible homogeneous of degree 2 in x1, x2, x3
# p = [p1, p2, p3] is a point on q.
#
# The output is a list X such that the substitution {x[1]=X[1], x[2]=X[2], x[3]=X[3]}
# turns q to the form x2^2 - x1*x3
#
Simplify_q := proc(q, x :: list, p :: list, X :: list) option trace;
  local x1, x2, x3, Q, i, pts, Q1, Q2, Q3, Line, Lines, EQ, v, SL, S;
  if nops(x) ≠ 3 or nops(p) ≠ 3 or p = [0, 0, 0] then
    error "wrong input"
  elif nargs = 3 then
    return procname(args, x)
  fi;
  x1, x2, x3 := op(x);
  if p[3] = 0 then
    return procname(q, x, [p[3], p[1], p[2]], subs({x1 = x2, x2 = x3, x3 = x1}, X))
  elif p[3] ≠ 1 then
    return procname(q, x, normal([p[1]/p[3], p[2]/p[3], 1]), X)
  elif p[1] ≠ 0 or p[2] ≠ 0 then
    return procname(q, x, [0, 0, 1], subs(x1 = x1 + p[1]*x3, x2 = x2 + p[2]*x3, X))
  fi;
  Q := collect(subs({x1 = X[1], x2 = X[2], x3 = X[3]}, q), {x1, x2, x3}, normal);
  if normal({seq(coeff(Q, x[i], 2), i = 1..3)}) ≠ {0} then
    pts := {[0, 0, 1]};
    # Now Q has the point [0,0,1] on it. Lets intersect Q with three lines through
    # that point to find more points, until we have three points.
    Q1 := normal(subs(x1 = 0, Q)/x2);
    if coeff(Q1, x3) ≠ 0 then
      pts := pts union {[0, 1, normal(-coeff(Q1, x2)/coeff(Q1, x3))]}
    fi;
    Q2 := normal(subs(x2 = 0, Q)/x1);
    if coeff(Q2, x3) ≠ 0 then
      pts := pts union {[1, 0, normal(-coeff(Q2, x1)/coeff(Q2, x3))]}
    fi;
    Q3 := normal(subs(x1 = x2, Q)/x2);
    if nops(pts) < 3 and coeff(Q3, x3) ≠ 0 then
      pts := pts union {[1, 1, normal(-coeff(Q3, x2)/coeff(Q3, x3))]}
    fi;
    if nops(pts) < 3 then
      error "Input must be irreducible, not a product of two lines"
    fi;
    Line := a1 * x1 + a2 * x2 + a3 * x3;
    Lines := {};
    for i in pts do
      EQ := {seq(eval(Line, {x1 = v[1], x2 = v[2], x3 = v[3]}), v = pts minus {i})};
      Lines := Lines union {primpart(subs(solve(EQ, {a1, a2, a3}), Line), {x1, x2,
x3})}

```

```

od;
SL := solve( {seq(y[i] = Lines[i], i = 1 ..3)}, {x1, x2, x3});
return procname(q, x, p, subs(seq(y[i] = x[i], i = 1 ..3), subs(SL, X) ) )
fi;
# Now [1,0,0], [0,1,0], [0,0,1] are on the curve.
# So we're looking at .. x1x2 + ..x1x3 + ..x2x3 and all of those coefficients
# are non-zero (otherwise the equation would be reducible).
# Aiming for x2^2 - x1 * x3
S := subs(x1 = x1 - x2 * coeff(coeff(Q, x2), x3) / coeff(coeff(Q, x1), x3), X);
Q := collect(subs( {x1 = S[1], x2 = S[2], x3 = S[3]}, q), {x1, x2, x3}, normal);
S := subs(x3 = -x3 * lcoeff(Q, x2) / coeff(coeff(Q, x3), x1), S);
Q := collect(subs( {x1 = S[1], x2 = S[2], x3 = S[3]}, q), {x1, x2, x3}, normal);
S := subs(x3 = x3 + x2 * coeff(coeff(Q, x2), x1) / lcoeff(Q, x2), S);
Q := primpart(collect(subs( {x1 = S[1], x2 = S[2], x3 = S[3]}, q), {x1, x2, x3}, normal),
x2);
if not member(x2^2 - x1 * x3, {Q, -Q}) then error "unexpected" fi;
S;
end:

```

> # Example:

> q := 3*x1^2 + x1*x2 + x1*x3 + x2^2 - 10*x2*x3 - 20*x3^2;

Simplify_q(q, [x1, x2, x3], [3, 2, 1]);

primpart(subs({x1 = %[1], x2 = %[2], x3 = %[3]}, q), x2);

$$q := 3x1^2 + x1x2 + x1x3 + x2^2 - 10x2x3 - 20x3^2$$

{--> enter Simplify_q, args = 3*x1^2+x1*x2+x1*x3+x2^2-10*x2*x3-20*x3^2, [x1, x2, x3], [3, 2, 1]}

{--> enter Simplify_q, args = 3*x1^2+x1*x2+x1*x3+x2^2-10*x2*x3-20*x3^2, [x1, x2, x3], [3, 2, 1], [x1, x2, x3]}

$$x1, x2, x3 := x1, x2, x3$$

{--> enter Simplify_q, args = 3*x1^2+x1*x2+x1*x3+x2^2-10*x2*x3-20*x3^2, [x1, x2, x3], [0, 0, 1], [x1+3*x3, x2+2*x3, x3]}

$$x1, x2, x3 := x1, x2, x3$$

$$Q := 3x1^2 + (21x3 + x2)x1 + x2^2 - 3x2x3$$

$$pts := \{[0, 0, 1]\}$$

$$Q1 := x2 - 3x3$$

$$pts := \left\{ [0, 0, 1], \left[0, 1, \frac{1}{3} \right] \right\}$$

$$Q2 := 3x1 + 21x3$$

$$pts := \left\{ [0, 0, 1], \left[0, 1, \frac{1}{3} \right], \left[1, 0, -\frac{1}{7} \right] \right\}$$

$$Q3 := 5x2 + 18x3$$

$$Line := a1x1 + a2x2 + a3x3$$

Lines := ∅

i := [0, 0, 1]

$$EQ := \left\{ a1 - \frac{a3}{7}, a2 + \frac{a3}{3} \right\}$$

Lines := {3 x1 - 7 x2 + 21 x3}

$$i := \left[0, 1, \frac{1}{3} \right]$$

$$EQ := \left\{ a3, a1 - \frac{a3}{7} \right\}$$

Lines := {x2, 3 x1 - 7 x2 + 21 x3}

$$i := \left[1, 0, -\frac{1}{7} \right]$$

$$EQ := \left\{ a3, a2 + \frac{a3}{3} \right\}$$

Lines := {x1, x2, 3 x1 - 7 x2 + 21 x3}

$$SL := \left\{ x1 = y_1, x2 = y_2, x3 = \frac{y_3}{21} - \frac{y_1}{7} + \frac{y_2}{3} \right\}$$

```
{--> enter Simplify_q, args = 3*x1^2+x1*x2+x1*x3+x2^2-10*x2*x3-20*
x3^2, [x1, x2, x3], [0, 0, 1], [(4/7)*x1+(1/7)*x3+x2, (5/3)*x2+(2/21)
*x3-(2/7)*x1, (1/21)*x3-(1/7)*x1+(1/3)*x2]
```

x1, x2, x3 := x1, x2, x3

$$Q := \left(x3 + \frac{59 x2}{7} \right) x1 - \frac{x2 x3}{7}$$

$$S := \left[\frac{4 x1}{7} + \frac{53 x2}{49} + \frac{x3}{7}, \frac{239 x2}{147} + \frac{2 x3}{21} - \frac{2 x1}{7}, \frac{x3}{21} - \frac{x1}{7} + \frac{46 x2}{147} \right]$$

$$Q := \left(x3 + \frac{59 x2}{7} \right) x1 + \frac{59 x2^2}{49}$$

$$S := \left[\frac{4 x1}{7} + \frac{53 x2}{49} - \frac{59 x3}{343}, \frac{239 x2}{147} - \frac{118 x3}{1029} - \frac{2 x1}{7}, -\frac{59 x3}{1029} - \frac{x1}{7} + \frac{46 x2}{147} \right]$$

$$Q := \left(\frac{59 x2}{7} - \frac{59 x3}{49} \right) x1 + \frac{59 x2^2}{49}$$

$$S := \left[\frac{4 x1}{7} - \frac{6 x2}{49} - \frac{59 x3}{343}, \frac{121 x2}{147} - \frac{118 x3}{1029} - \frac{2 x1}{7}, -\frac{59 x3}{1029} - \frac{13 x2}{147} - \frac{x1}{7} \right]$$

$$Q := -x1 x3 + x2^2$$

$$\left[\frac{4 x1}{7} - \frac{6 x2}{49} - \frac{59 x3}{343}, \frac{121 x2}{147} - \frac{118 x3}{1029} - \frac{2 x1}{7}, -\frac{59 x3}{1029} - \frac{13 x2}{147} - \frac{x1}{7} \right]$$

```
<-- exit Simplify_q (now in Simplify_q) = [(4/7)*x1-(6/49)*x2-
(59/343)*x3, (121/147)*x2-(118/1029)*x3-(2/7)*x1, -(59/1029)*x3-
(13/147)*x2-(1/7)*x1]
```

```
<-- exit Simplify_q (now in Simplify_q) = [(4/7)*x1-(6/49)*x2-
(59/343)*x3, (121/147)*x2-(118/1029)*x3-(2/7)*x1, -(59/1029)*x3-
```

```

(13/147)*x2-(1/7)*x1]}
<-- exit Simplify_q (now in Simplify_q) = [(4/7)*x1-(6/49)*x2-
(59/343)*x3, (121/147)*x2-(118/1029)*x3-(2/7)*x1, -(59/1029)*x3-
(13/147)*x2-(1/7)*x1]}
<-- exit Simplify_q (now at top level) = [(4/7)*x1-(6/49)*x2-
(59/343)*x3, (121/147)*x2-(118/1029)*x3-(2/7)*x1, -(59/1029)*x3-
(13/147)*x2-(1/7)*x1]}

```

$$\left[\frac{4x1}{7} - \frac{6x2}{49} - \frac{59x3}{343}, \frac{121x2}{147} - \frac{118x3}{1029} - \frac{2x1}{7}, -\frac{59x3}{1029} - \frac{13x2}{147} - \frac{x1}{7} \right]$$

$$-x1x3 + x2^2$$

(1)

